

estimates table — Compare estimation results

[Description](#)
[Options](#)
[Also see](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Stored results](#)

[Syntax](#)
[References](#)

Description

`estimates table` organizes estimation results from one or more models in a single formatted table.

Quick start

Display a table of coefficients for stored estimates `m1` and `m2`

```
estimates table m1 m2
```

Same as above, but include standard errors

```
estimates table m1 m2, se
```

Same as above, but display coefficients and standard errors to 2 decimal places

```
estimates table m1 m2, b(%7.2f) se(%7.2f)
```

Same as above, but include p -values displayed to 3 decimal places

```
estimates table m1 m2, b(%7.2f) se(%7.2f) p(%4.3f)
```

Table of coefficients for `m1` and `m2` with sample size and adjusted R^2

```
estimates table m1 m2, stats(N r2_a)
```

Same as above, but replace variable names with labels

```
estimates table m1 m2, stats(N r2_a) varlabel
```

Table of coefficients with stars to denote significance

```
estimates table m1 m2, star
```

Display coefficients in exponentiated form

```
estimates table m3 m4, eform
```

Display only a subset of variables and reorder variables in table

```
estimates table m1 m2, keep(v2 v1 v3 _cons)
```

Menu

Statistics > Postestimation

Syntax

```
estimates table [ namelist ] [ , options ]
```

namelist is the name given to previously stored estimation results, a list of names, `_all`, or `*`. A name may be `.`, meaning the current (active) estimates. `_all` and `*` mean the same thing.

| <i>options</i> | Description |
|----------------|-------------|
|----------------|-------------|

Main

| | |
|---------------------------------------|-----------------------------------|
| <code>stats(<i>scalarlist</i>)</code> | report <i>scalarlist</i> in table |
| <code>star[(#1 #2 #3)]</code> | use stars to denote significance |

Options

| | |
|--|--|
| <code>keep(<i>coeflist</i>)</code> | report coefficients in order specified |
| <code>drop(<i>coeflist</i>)</code> | omit specified coefficients from table |
| <code>equations(<i>matchlist</i>)</code> | match equations of models as specified |

Numerical formats

| | |
|---------------------------|---|
| <code>b[(%fmt)]</code> | how to format coefficients, which are always reported |
| <code>se[(%fmt)]</code> | report standard errors and use optional format |
| <code>t[(%fmt)]</code> | report <i>t</i> or <i>z</i> and use optional format |
| <code>p[(%fmt)]</code> | report <i>p</i> -values and use optional format |
| <code>stfmt(%fmt)</code> | how to format scalar statistics |

General format

| | |
|-----------------------------|---|
| <code>varwidth(#)</code> | use # characters to display variable names and statistics |
| <code>modelwidth(#)</code> | use # characters to display model names |
| <code>eform</code> | display coefficients in exponentiated form |
| <code>varlabel</code> | display variable labels rather than variable names |
| <code>newpanel</code> | display statistics in separate table from coefficients |
| <code>style(online)</code> | put vertical line after variable names; the default |
| <code>style(columns)</code> | put vertical line separating every column |
| <code>style(noline)</code> | suppress all vertical lines |
| <code>coded</code> | display compact table |

Reporting

| | |
|------------------------------|--|
| <code>display_options</code> | control row spacing, line width, and display of omitted variables and base and empty cells |
| <code>title(string)</code> | title for table |

collect is allowed; see [U] 11.1.10 Prefix commands.

title() does not appear in the dialog box.

Options

Main

`stats(scalarlist)` specifies a list of any of or all the names of scalars stored in `e()` to be displayed in the table. *scalarlist* may also contain the following:

| | |
|-------------------|---|
| <code>aic</code> | Akaike's information criterion |
| <code>bic</code> | Schwarz's Bayesian information criterion |
| <code>rank</code> | rank of <code>e(V)</code> (# of free parameters in model) |

The specified statistics do not have to be available for all estimation results being displayed.

For example, `stats(N l1 chi2 aic)` specifies that `e(N)`, `e(l1)`, `e(chi2)`, and AIC be included. In Stata, `e(N)` records the number of observations; `e(l1)`, the log likelihood; and `e(chi2)`, the χ^2 test that all coefficients in the first equation of the model are equal to zero.

`star` and `star(#1 #2 #3)` specify that stars (asterisks) are to be used to mark significance. The second syntax specifies the significance for one, two, and three stars. If you specify simply `star`, that is equivalent to specifying `star(.05 .01 .001)`, which means one star (*) if $p < 0.05$, two stars (**) if $p < 0.01$, and three stars (***) if $p < 0.001$.

The `star` and `star()` options may not be combined with the `se`, `t`, or `p` option.

Options

`keep(coeflist)` and `drop(coeflist)` are alternatives; they specify coefficients to be included or omitted from the table. The default is to display all coefficients.

If `keep()` is specified, it specifies not only the coefficients to be included but also the order in which they appear.

A *coeflist* is a list of coefficient names, each name of which may be simple (for example, `price`), an equation name followed by a colon (for example, `mean:`), or a full name (for example, `mean:price`). Names are separated from each other by blanks.

When full names are not specified, all coefficients that match the partial specification are included. For instance, `drop(_cons)` would omit `_cons` for all equations.

`equations(matchlist)` specifies how the equations of the models in *namelist* are to be matched. The default is to match equations by name. Matching by name usually works well when all results were fit by the same estimation command. When you are comparing results from different estimation commands, however, specifying `equations()` may be necessary.

The most common usage is `equations(1)`, which indicates that all first equations are to be matched into one equation named `#1`.

matchlist has the syntax

```
term [ , term ... ]
```

term is

```
[eqname = ] #:#. . . :#           (syntax 1)
```

```
[eqname = ] #                   (syntax 2)
```

In syntax 1, each `#` is a number or a period (`.`). If a number, it specifies the position of the equation in the corresponding model; `1:3:1` would indicate that equation 1 in the first model matches equation 3 in the second, which matches equation 1 in the third. A period indicates that there

is no corresponding equation in the model; 1::1 indicates that equation 1 in the first matches equation 1 in the third.

In syntax 2, you specify just one number, say, 1 or 2, and that is shorthand for 1:1...:1 or 2:2...:2, meaning that equation 1 matches across all models specified or that equation 2 matches across all models specified.

Now that you can specify a *term*, you can put that together into a *matchlist* by separating one term from the other by commas. In what follows, we will assume that three names were specified,

```
. estimates table alpha beta gamma, ...
```

`equations(1)` is equivalent to `equations(1:1:1)`; we would be saying that the first equations match across the board.

`equations(1::1)` would specify that equation 1 matches in models `alpha` and `gamma` but that there is nothing corresponding in model `beta`.

`equations(1,2)` is equivalent to `equations(1:1:1, 2:2:2)`. We would be saying that the first equations match across the board and so do the second equations.

`equations(1, 2::2)` would specify that the first equations match across the board, that the second equations match for models `alpha` and `gamma`, and that there is nothing equivalent to equation 2 in model `beta`.

If `equations()` is specified, equations not matched by position are matched by name.

Numerical formats

`b(%fmt)` specifies how the coefficients are to be displayed. You might specify `b(%9.2f)` to make decimal points line up. There is also a `b` option, which specifies that coefficients are to be displayed, but that is just included for consistency with the `se`, `t`, and `p` options. Coefficients are always displayed.

`se`, `t`, and `p` specify that standard errors, t or z statistics, and p -values are to be displayed. The default is not to display them. `se(%fmt)`, `t(%fmt)`, and `p(%fmt)` specify that each is to be displayed and specifies the display format to be used to format them.

`stfmt(%fmt)` specifies the format for displaying the scalar statistics included by the `stats()` option.

General format

`varwidth(#)` specifies the number of character positions used to display the names of the variables and statistics. The default is 12.

`modelwidth(#)` specifies the number of character positions used to display the names of the models. The default is 12.

`eform` displays coefficients in exponentiated form. For each coefficient, $\exp(\beta)$ rather than β is displayed, and standard errors are transformed appropriately. Display of the intercept, if any, is suppressed.

`varlabel` specifies that variable labels be displayed instead of variable names.

`newpanel` specifies that the statistics be displayed in a table separated by a blank line from the table with coefficients rather than in the style of another equation in the table of coefficients.

`style(stylespec)` specifies the style of the coefficient table.

`style(online)` specifies that a vertical line be displayed after the variables but not between the models. This is the default.

`style(columns)` specifies that vertical lines be displayed after each column.

`style(noline)` specifies that no vertical lines be displayed.

`coded` specifies that a compact table be displayed. This format is especially useful for comparing variables that are included in a large collection of models.

Reporting

display_options: `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, and `fvwrapon(style)`; see [R] [Estimation options](#).

The following option is available with `estimates table` but is not shown in the dialog box:

`title(string)` specifies the title to appear above the table.

Remarks and examples

[stata.com](http://www.stata.com)

`estimates table` lets you format estimation results and organize results from multiple models in a single table. You achieve this by combining `estimates table` with `estimates store`; see [R] [estimates store](#). When combined with `putdocx` or `putpdf`, `estimates table` lets you create customized tables of results in `.docx` or `.pdf` format.

Alternatively, you can use `etable` to create a table with the results stored with `estimates store`. The advantage of using `etable` is that you can add notes to your table and export it to a variety of file types, such as HTML and L^AT_EX.

► Example 1: Creating the default table

If you type `estimates table` without arguments, a table of the most recent estimation results will be shown:

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. regress mpg weight
(output omitted)
. regress mpg weight displ
(output omitted)
. estimates table
```

| Variable | Active |
|--------------|------------|
| weight | -.00656711 |
| displacement | .00528078 |
| _cons | 40.084522 |

In the above output table, only the results from the second `regress` command are displayed because they are the current (active) estimates.

◀

By default, `estimates table` displays only coefficient estimates. You can request additional model statistics by specifying the `stats()` option. Estimation commands store e-class results. To see a list of available results from the last estimation command that can be specified in the `stats()` option, you can use the [ereturn list](#) command.

▶ Example 2: Creating a table with multiple models

If we want to display more than one set of estimation results in a single table, we can use `estimates store` to store each set of estimation results in memory so that they can be accessed later by `estimates table`. This is one of the primary uses of `estimates table`—comparing estimation results:

```
. regress mpg weight displ
  (output omitted)
. estimates store base
. regress mpg weight displ i.foreign
  (output omitted)
. estimates store alt
. qreg mpg weight displ i.foreign
  (output omitted)
. estimates store qreg
. estimates table base alt qreg, stats(r2)
```

| Variable | base | alt | qreg |
|--------------|------------|------------|------------|
| weight | -.00656711 | -.00677449 | -.00595056 |
| displacement | .00528078 | .00192865 | .00018552 |
| foreign | | | |
| Foreign | | -1.6006312 | -2.1326005 |
| _cons | 40.084522 | 41.847949 | 39.213348 |
| r2 | .6529307 | .66287957 | |

`estimates table` automatically lines up the point estimates of coefficients on covariates that are common across each model. The `stats(r2)` option specifies that the coefficient of determination (R^2) be placed under the models for which it is computed.

The `foreign` variable was included in the models `alt` and `qreg` as a factor variable. Because the values of `foreign` are labeled, the value labels are displayed by default. See [\[D\] label](#) for information about managing value labels.

◀

▶ Example 3: Creating and exporting a formatted table

We can add estimates of the standard errors to the table and format them along with the original coefficient estimates by using the `b()` and `se()` options.

As shown in [example 2](#), the labeled values of any factor variables are displayed by default. To see the labels corresponding to the variables that are included in the model rather than the names of the variables, we can specify the `varlabel` option.

```
. estimates table alt, b(%5.4f) se(%5.4f) varlabel
```


| Variable | alt |
|------------------------|-------------------|
| Weight (lbs.) | -0.0068 0.0012 |
| Displacement (cu. in.) | 0.0019 0.0101 |
| Car origin | |
| Foreign | -1.6006 1.1136 |
| Constant | 41.8479 2.3507 |

Legend: b/se

If we wanted this table to appear in a Word document, we could use the `putdocx` command with the `etable` output type to write it to a new document. To create the new document `myresults.docx`, we would type the following `putdocx` commands:

```
. putdocx begin
. putdocx table results = etable
. putdocx save myresults.docx
successfully created "myresults.docx"
```

This creates a table in Word that looks like



The screenshot shows a Microsoft Word document titled "myresults [Compatibility Mode] - Word". The ribbon is set to "TABLE TOOLS" > "DESIGN" > "LAYOUT". The table content is as follows:

| Variable | alt |
|------------------------|-------------------|
| Weight (lbs.) | -0.0068 0.0012 |
| Displacement (cu. in.) | 0.0019 0.0101 |
| Car origin | |
| Foreign | -1.6006 1.1136 |
| Constant | 41.8479 2.3507 |

The status bar at the bottom indicates "PAGE 1 OF 1", "19 WORDS", and "ENGLISH (UNITED STATES)".

Stored results

`estimates table` stores the following in `r()`:

Macros

`r(names)` names of results used

Matrices

`r(coef)` matrix $M: n \times 2*m$
 $M[i, 2j-1]$ = i th parameter estimate for model j ;
 $M[i, 2j]$ = variance of $M[i, 2j-1]$; $i=1, \dots, n$; $j=1, \dots, m$

`r(stats)` matrix $S: k \times m$ (if option `stats()` specified)
 $S[i, j]$ = i th statistic for model j ; $i=1, \dots, k$; $j=1, \dots, m$

References

Gallup, J. L. 2012. A new system for formatting estimation tables. *Stata Journal* 12: 3–28.

Weiss, M. 2010. *Stata tip 90: Displaying partial results*. *Stata Journal* 10: 500–502.

Also see

[R] [estimates](#) — Save and manipulate estimation results

[R] [etable](#) — Create a table of estimation results

[R] [table regression](#) — Table of regression results

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the [FAQ on citing Stata documentation](#).